**Table 1 Pharmacokinetic and survival approaches describing the time course of drug (initial amount 100 mass units or 100 molecules; vmax=100; km=50; k=vmax/km)**

| | First-order | | Mixed-Order | |
|---|---|---|---|---|
| Time | Amount | Survival | Amount | Survival |
| 0 | 100.000 | 100.000 | 100.000 | 100.000 |
| 1 | 13.534 | 13.534 | 42.630 | 42.630 |
| 2 | 1.832 | 1.832 | 10.886 | 10.886 |
| 3 | 0.248 | 0.248 | 1.768 | 1.768 |
| 4 | 0.034 | 0.034 | 0.247 | 0.247 |
| 5 | 0.005 | 0.005 | 0.034 | 0.034 |

**Table 2 The exponential family of hazard functions and corresponding named distributions**

| Hazard | Distribution | |
|---|---|---|
| $h = \lambda_0 \times e^0$ | Exponential | **Equation 17** |
| $h(t) = \lambda_0 \times e^{\beta \times t}$ | Gompertz | **Equation 18** |
| $h(t) = \lambda_0 \times e^{\beta \times \ln(t)}$ | Weibull | **Equation 19** |
| $h(X_1, X_2, \ldots) = \lambda_0 \times e^{\beta_1 \times X_1 + \beta_2 \times X_2 + \cdots}$ | Generalized exponential hazard | **Equation 20** |

**Table 3 Likelihood of an event**

| Observation | Likelihood | Event Time |
|---|---|---|
| Exact time | $S(t) \times h(t)$ | Event occurs at time $t$ |
| Interval censored | $S(t_{start}) - S(t_{end})$ | Event occurs between $t_{start}$ and $t_{end}$ |
| Right censored | $S(t)$ | Event has not occurred at time $t$ |

```
Berkeley Madonna Code Illustrating Constant, Gompertz and
Weibull Hazard Functions

METHOD stiff

STARTTIME = 0
STOPTIME=100
DT = 0.001
DTOUT=1

iswb=0 ; use Weibull hazard
isgm=0 ; use Gompertz hazard
istci=1 ; use doses to reach target conc
isload=0 ; use a loading dose

;PK model parameters
dose=if istci then 0 else 100
target=20 ; target concentration
duration=52 ; duration of treatment
tabs=1 ; absorption half-life
cl=0.3 ; clearance
v=1  ; volume of distribution

;Hazard function parameters
base=if iswb then 0.01 else if isgm then 0.01 else  0.0625
betawb=if iswb then 0.51 else 0
betagm=if isgm then 0.037 else 0
betacp=if iswb then -0.025 else if isgm then -0.025 else  -0.025

;PK model variables
ka=logn(2)/tabs
gutratein=ka*gut
ldose=if isload then if istci then v*target else 0 else 0
mdr=if istci then cl*target*squarepulse(0,duration) else 0
init gut=dose
init amt=ldose
init cumhaztrt=0
init cumhazpla=0
init cumRR=0

d/dt(gut)=-gutratein
d/dt(amt)=gutratein + mdr - cl*cp
d/dt(cumhaztrt)=haztrt
d/dt(cumhazpla)=hazpla
d/dt(cumRR)=relrisk
```

```
cp=amt/v
consthaz=base
wbhaz=if (time=0) then 0 else exp(betawb*logn(time))
gmhaz=exp(betagm*time)

usf=base*wbhaz*gmhaz
haztrt=usf*(1+betacp*cp)
hazpla=usf

survtrt=exp(-cumhaztrt)
survpla=exp(-cumhazpla)

pdftrt=survtrt*haztrt
pdfpla=survpla*hazpla

relrisk=if cumhazpla=0 then 1 else cumhaztrt/cumhazpla
avgRR=if time=0 then 1 else cumRR/time
```

Berkeley Madonna Code Illustrating Parkinson's Disease Hazard of Mortality and Dropout with Different Treatments [1]

```
METHOD RK4

STARTTIME = 0
STOPTIME=8
DT = 0.02
RENAME time=year
DTOUT=0.1

;disease status model
updrststatus=1
pigdstatus=2

;hazard models
indephaz=0
cr230=2 ; Final model code

;event models
event_DEA=0
event_DRP=1

;disease status model choice
typestatus=1 ;updrststatus

;hazard model choice
hazmdl=0 ; indephaz=0 (death model), cr230=2 (dropout model)

;event model choice
event=0 ;event_DEA=0, event_DRP=1

;treatment choice codes
ld=1 ; levodopa
dp=2 ; selegiline
lddp=3 ; levodopa + selegiline

;assign treatment and selegiline treatment delay
trt=3
dplag=2

;miscellaneous choices
iscr=0 ; competing risk parameters
isPIGD=0; baseline PIGD
```

```
;Model Parameters
;UPDRS total disease progress
popS0[updrststatus]=20.9 ; S0 UPDRS
popRSS[updrststatus]=85.2 ; Rss UPDRS
popTP[updrststatus]=109 ; y
;Drug effects on half-time of progress
popkplt[updrststatus]=0.497 ; levodopa
popkpdt[updrststatus]=0.357 ; deprenyl
popkpbt[updrststatus]=1.11 ; bromocriptine
popkppt[updrststatus]=-0.143 ; pergolide
popkptt[updrststatus]=0 ; tocopherol
popflxdt[updrststatus]=3.76 ; LD and DP interaction

;Offset effects of levodopa and other treatments
popem0L[updrststatus]=0 ; u
popc5L[updrststatus]=0.0291 ; 0.3g/d
popBEML[updrststatus]= -24.4 ; u
popTEML[updrststatus]=0.606 ; y
popTEQL[updrststatus]=2.82 ; y
popHill[updrststatus]=0.551
popBOD[updrststatus]= -0.401 ; u
popBOB[updrststatus]= -1.56 ; u
popBOP[updrststatus]= -0.692 ; u

;PIGD disease progress
;estimates from ??
popS0[pigdstatus]=1.51 ; u
popRSS[pigdstatus]=17.6 ; u
popTP[pigdstatus]=40.1 ; y

;Drug effects on half-time of progress
popkplt[pigdstatus]=0.659 ; levodopa
popkpdt[pigdstatus]=0.407 ; deprenyl
popkpbt[pigdstatus]=-1.18 ; bromocriptine
popkppt[pigdstatus]=0.378 ; pergolide
popkptt[pigdstatus]=0 ; tocopherol
popflxdt[pigdstatus]=5.41 ; LD and DP interaction

;Offset effects of levodopa and other treatments
popem0L[pigdstatus]=0
popc5L[pigdstatus]=4.07
popBEML[pigdstatus]=-2.93
popTEML[pigdstatus]=0.364
popTEQL[pigdstatus]=0.525
popHill[pigdstatus]=0.51
popBOD[pigdstatus]=-0.228
popBOB[pigdstatus]=-0.499
```

```
popBOP[pigdstatus]=-0.0826

; Assign model specific parameters
S0=popS0[typestatus]
RSS=popRSS[typestatus]
TP=popTP[typestatus]
kplt=popkplt[typestatus]
kpdt=popkpdt[typestatus]
flxdt=popflxdt[typestatus]
kpbt=popkpbt[typestatus]
kppt=popkppt[typestatus]
kptt=popkptt[typestatus]
em0L=popem0L[typestatus]
c5L=popc5L[typestatus]
BEML=popBEML[typestatus]
TEML=popTEML[typestatus]
TEQL=popTEQL[typestatus]
Hill=popHill[typestatus]
BOD=popBOD[typestatus]
BOB=popBOB[typestatus]
BOP=popBOP[typestatus]

isld=if (trt=ld or trt=lddp) then 1 else 0
isdp=if (trt=dp or trt=lddp) then 1 else 0
isdplag=if (year>dplag+0.1) then 1 else 0

;Treatment relative to median daily dose
CPL=isld ; 1=300 mg/day levodopa
CPD=isdp*isdplag ; 1=10 mg/day deprenyl
CPB=0 ; 1=15 mg/day bromocriptine
CPP=0 ; 1=1.25 mg/day pergolide
CPT=0 ; 1= tocopherol

age0=60
age=age0+year

;Hazard function for event
;Independent Risks for death
popBASEGM[indephaz,event_DEA]=1.28 * 1e-5
popBTIME[indephaz,event_DEA]=0.127
popBAGE[indephaz,event_DEA]=0.0649
popBETASTATUS[indephaz,event_DEA]=0.0336
popBETADP[indephaz,event_DEA]=0.931

popBASEGM2[indephaz,event_DEA]=0
popBTIM2[indephaz,event_DEA]=0
popBLNTIM[indephaz,event_DEA]=0
```

```
popBETALD[indephaz,event_DEA]=0
popBETAPIGD[indephaz,event_DEA]=0
popBASEWB[indephaz,event_DEA]=0

;Dropout model parameters for death events as a cause of dropout

popBASEGM[cr230,event_DEA]=0.861*1e-9
popBTIME[cr230,event_DEA]=2.86
popBASEGM2[cr230,event_DEA]=0
popBTIM2[cr230,event_DEA]=0
popBETALD[cr230,event_DEA]=0
popBETADP[cr230,event_DEA]=0.581
popBETASTATUS[cr230,event_DEA]=0.00826
popBETAPIGD[cr230,event_DEA]=0
popBASEWB[cr230,event_DEA]=35.9*1e-3
popBLNTIM[cr230,event_DEA]=1.35
popBAGE[cr230,event_DEA]=0

;Dropout model parameters for non-death events as a cause of
dropout

popBASEGM[cr230,event_DRP]=76.4*1e-3
popBTIME[cr230,event_DRP]=-0.417
popBASEGM2[cr230,event_DRP]=0.0525*1e-9
popBTIM2[cr230,event_DRP]=3.4
popBETALD[cr230,event_DRP]=0
popBETADP[cr230,event_DRP]=-1.6
popBETASTATUS[cr230,event_DRP]=0.0391
popBETAPIGD[cr230,event_DRP]=0
popBASEWB[cr230,event_DRP]=0.303*1e-3
popBLNTIM[cr230,event_DRP]=2.99
popBAGE[cr230,event_DRP]=0

;Assign hazard parameters
basegm=popBASEGM[hazmdl,event] ; baseline events/year
btime=popBTIME[hazmdl,event] ; 1/year
basegm2=popBASEGM2[hazmdl,event] ; baseline events/year
btime2=popBTIM2[hazmdl,event] ; 1/year
betald=popBETALD[hazmdl,event] ; 1/300 mg/day
betadp=popBETADP[hazmdl,event] ; 1/10 mg/day
betastatus=popBETASTATUS[hazmdl,event] ; 1/UPDRS
betaPIGD=popBETAPIGD[hazmdl,event] ; Baseline subtype
basewb=popBASEWB[hazmdl,event]
blntim=popBLNTIM[hazmdl,event] ; 1/ln(year)
bage=popBAGE[hazmdl,event] ; 1/year

piCR=if iscr then 0.122 else 1 ; competing risk prob of event
```

```
;Model Variables

fplt=exp(CPL*kplt)
fpdt=exp (CPD*kpdt)
fpt=exp(CPT*kptt)
fpp=exp(CPP*kppt)
fpb=exp(CPB*kpbt)

FPLD=if (CPL> 0 and CPD>0) then flxdt*fplt*fpdt else if CPL>0
then fplt else if CPD>0 then fpdt else 1

EML=EM0L + BEML*(1-exp(-logn(2)/TEML*year))
Offset=EML*CeL^Hill/(C5L^Hill + CeL^Hill) + CPD*BOD + CPB*BOB +
CPP*BOP

TrtStatus=Offset + Treated
PlaStatus=NatHist

lntime=if year<=0 then 1e-10 else logn(year)
gompertz=btime*year
hage=bage*age
hld=if CPL>0 then CeL*betald else 0
hdp=if CPD>0 then betadp else 0
hPIGD=if isPIGD then betaPIGD else 0
hgm=btime*year+hage+hPIGD
hgm2=btime2*year
hwb=blntim*lntime

isgm1=1
isgm2=1
iswb=if year<=0 then 0 else 1
HazPla=isgm1*basegm*exp(hgm+betastatus*PlaStatus) +
isgm2*basegm2*exp(hgm2)+ iswb*basewb*exp(hwb)
HazTrt=isgm1*basegm*exp(hgm+betastatus*TrtStatus+hld+hdp)+
isgm2*basegm2*exp(hgm2)+ iswb*basewb*exp(hwb)

init NatHist=S0
init Treated=S0
init CeL=0
init CumHazTrt=0
Init CumHazPla=0

d/dt(NatHist)=logn(2)/TP*(RSS-NatHist)*NatHist
d/dt(Treated)=logn(2)/(TP*FPLD*fpt*fpp*fpb)*(RSS-
Treated)*Treated
d/dt(CeL)=logn(2)/TEQL*(CPL - CeL)
```

```
d/dt(CumHazTrt)=HazTrt
d/dt(CumHazPla)=HazPla

probSurvivalTrt=piCR*exp(-CumHazTrt) + 1-piCR
probSurvivalPla=piCR*exp(-CumHazPla) + 1-piCR
probdensityTrt=(1-probSurvivalTrt)*HazTrt
probdensityPla=(1-probSurvivalPla)*HazPla

NCumHazTrt=100*CumHazTrt
NCumHazPla=100*CumHazPla
RelRisk=if NCumHazPla>0 then NCumHazTrt/NCumHazPla else 1
```

1. Vu, T.C., J.G. Nutt, and N.H.G. Holford, *Disease progress and response to treatment as predictors of survival, disability, cognitive impairment and depression in Parkinson's disease.* British Journal of Clinical Pharmacology, 2012. **74**(2): p. 284-295.